// File: noderental.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";

/**
 * @title KaspaNodes Rental Contract
 * @dev This contract enables users to rent different types of KaspaNodes in exchange for KASNODE tokens.
 * Each node type has a specific rental price, payout amount, and rental period.
 * Users can claim their payout at the end of the rental period.
 * The contract owner can update node parameters and recover ERC20 tokens sent by mistake.
 */
contract NodeRental is Ownable {
    IERC20 public kasnodeToken;
    uint256 public constant TOTAL_SUPPLY = 28700000000 * (10 ** 18);

    struct Node {
        uint256 rentalPrice;
        uint256 payoutAmount;
        uint256 rentalPeriod;
        string nodeType;
    }

    Node public rapidNode;
    Node public ghostNode;
    Node public heavyNode;

    mapping(address => Node) public rentedNodes;
    mapping(address => uint256) public rentalStartTime;

    event NodeRented(address indexed renter, string nodeType, uint256 amount);
    event PayoutClaimed(address indexed renter, string nodeType, uint256 amount);

    bool public emergencyStopped = false;

    modifier stopInEmergency {
        require(!emergencyStopped, "Contract is stopped in an emergency");
        _;
    }

    constructor(address _kasnodeTokenAddress) Ownable(msg.sender) {
        kasnodeToken = IERC20(_kasnodeTokenAddress);
        rapidNode = Node(TOTAL_SUPPLY / 1000, 50 * (10 ** 18), 30 days, "RapidNode");
        ghostNode = Node(TOTAL_SUPPLY / 100, 500 * (10 ** 18), 90 days, "GhostNode");
        heavyNode = Node(TOTAL_SUPPLY / 20, 2500 * (10 ** 18), 180 days, "HeavyNode");
    }

    function rentRapidNode() external stopInEmergency {
        rentNode(rapidNode);
    }

    function rentGhostNode() external stopInEmergency {
        rentNode(ghostNode);
    }

    function rentHeavyNode() external stopInEmergency {
        rentNode(heavyNode);
    }

    function rentNode(Node memory node) internal {
        uint256 allowance = kasnodeToken.allowance(msg.sender, address(this));
        require(allowance >= node.rentalPrice, "Insufficient token allowance for node rental.");
        require(rentalStartTime[msg.sender] == 0 || rentalStartTime[msg.sender] + rentedNodes[msg.sender].rentalPeriod < block.timestamp, "Current node rental is still active or not yet started.");

        kasnodeToken.transferFrom(msg.sender, owner(), node.rentalPrice);
        rentedNodes[msg.sender] = node;
        rentalStartTime[msg.sender] = block.timestamp;

        emit NodeRented(msg.sender, node.nodeType, node.rentalPrice);
    }

    function claimPayout() external stopInEmergency {
        Node memory userNode = rentedNodes[msg.sender];
        require(userNode.rentalPrice > 0, "You have not rented a node.");
        require(rentalStartTime[msg.sender] + userNode.rentalPeriod <= block.timestamp, "Node rental period has not yet ended.");

        uint256 payoutAmount = userNode.payoutAmount;
        require(payoutAmount > 0, "No payout available.");
        require(kasnodeToken.balanceOf(address(this)) >= payoutAmount, "Insufficient funds in the contract for payout.");

        kasnodeToken.transfer(msg.sender, payoutAmount);
        rentalStartTime[msg.sender] = 0;

        emit PayoutClaimed(msg.sender, userNode.nodeType, payoutAmount);
    }

    function updateNodeParameters(string memory nodeType, uint256 rentalPrice, uint256 payoutAmount, uint256 rentalPeriod) external onlyOwner {
        if (keccak256(bytes(nodeType)) == keccak256(bytes("RapidNode"))) {
            rapidNode.rentalPrice = rentalPrice;
            rapidNode.payoutAmount = payoutAmount;
            rapidNode.rentalPeriod = rentalPeriod;
        } else if (keccak256(bytes(nodeType)) == keccak256(bytes("GhostNode"))) {
            ghostNode.rentalPrice = rentalPrice;
            ghostNode.payoutAmount = payoutAmount;
            ghostNode.rentalPeriod = rentalPeriod;
        } else if (keccak256(bytes(nodeType)) == keccak256(bytes("HeavyNode"))) {
            heavyNode.rentalPrice = rentalPrice;
            heavyNode.payoutAmount = payoutAmount;
            heavyNode.rentalPeriod = rentalPeriod;
        }
    }

    function toggleEmergencyStop() external onlyOwner {
        emergencyStopped = !emergencyStopped;
    }

    function recoverERC20(address tokenAddress, uint256 tokenAmount) external onlyOwner {
        require(tokenAddress != address(kasnodeToken), "Cannot recover KASNODE tokens");
        IERC20(tokenAddress).transfer(owner(), tokenAmount);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.20;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the value of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the value of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves a `value` amount of tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 value) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets a `value` amount of tokens as the allowance of `spender` over the
     * caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 value) external returns (bool);

    /**
     * @dev Moves a `value` amount of tokens from `from` to `to` using the
     * allowance mechanism. `value` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 value) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}